Pensieve header: Formal Gaussian integration over an arbitrary “Feynman Ring”; last to use Picard iteration.

What must a Feynman Ring F have? (Over some set of labels $\mathcal{S}$ with elements x, y, ...)
* A vector space over $\mathbb{Q}$.
* Has a symmetric linear $Z \longmapsto \partial_{x,y} Z$ and a symmetric bilinear $(Z_1, Z_2) \longmapsto (\partial_x Z_1)(\partial_y Z_2)$ that satisfy the axioms of (roughly) a connected circuit algebra.
* Has $q_{x,y} : F \to \mathbb{Q}$ in some sense dual to some $\theta_{x,y} \in F$.
* Has $\text{Ev}_{\text{vs}\to 0} : F \to F$.

Further axioms must be worked out.

Goals.
* Define $\int$.
* Prove a Fubini theorem.
* Prove a theorem about the injectivity of the Laplace transform.

Initialization

In[∘]:=
```
CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
CCF[ℰ_] := Factor[ℰ];
CF[ω_. ℰ_𝔼] := CF[ω] × CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p)__, ∞] ⋃ {x, p, ϵ}, ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :→ CCF[c] (Times @@ vs^ps)] ];
```

The Basic Feynman Ring

In[∘]:=
```
𝒮 = {x, x_, y, z};
q_{x_,y_}[f_] := (∂_{x,y} f) /. Thread[𝒮 → 0];
θ_{x_,y_} := x y;
f_ ≡ 0 := f === 0;
Ev_{vs_List→0}[f_] := CF[f /. Thread[vs → 0]]
```

The ϵSeries Feynman Ring

In[∘]:=
```
𝒮 = {x, y, z, ϕ, x_, p_, x̲_, p̲_};
q_{x_,y_}[ser_ϵSeries] := (∂_{x,y} ser⟦1⟧) /. Thread[𝒮 → 0];
θ_{x_,y_} := x y;
ϵSeries /: D[ser_ϵSeries, vs___] := D[#, vs] & /@ ser;
ϵSeries /: s1_ϵSeries + s2_ϵSeries :=
  ϵSeries @@ Table[s1⟦kk⟧ + s2⟦kk⟧, {kk, Min[Length@s1, Length@s2]}];
ϵSeries /: t_ + ser_ϵSeries := MapAt[(# + t) &, ser, 1];
ϵSeries /: s1_ϵSeries * s2_ϵSeries := ϵSeries @@ Table[
    Sum[s1⟦ii + 1⟧ × s2⟦kk - ii + 1⟧, {ii, 0, kk}], {kk, 0, Min[Length@s1, Length@s2] - 1}];
ϵSeries /: c_ * ser_ϵSeries := (c #) & /@ ser;
ser_ϵSeries ≡ 0 := And @@ ((# === 0) & /@ ser);
ϵSeries /: Integrate[ser_ϵSeries, pars__] := ϵSeries @@ (Integrate[#, pars] & /@ ser);
ϵSeries /: Ev_{vs_List→0}[ser_ϵSeries] := ser /. Thread[vs → 0];
CF[ser_ϵSeries] := CF /@ ser;
```

Integration

Using Picard Iteration!

In[∘]:=
```
𝔼 /: 𝔼[A_] × 𝔼[B_] := 𝔼[A + B]
```

In[∘]:=
```
𝔼[sd_SeriesData] /; (List @@ sd)⟦{1, 2, 4, 6}⟧ === {ϵ, 0, 0, 1} :=
 𝔼[ϵSeries @@ PadRight[sd⟦3⟧, sd⟦5⟧, 0]]
```

In[∘]:=

```
Unprotect[Integrate];
Integrate::sing = "How dare you ask me to integrate a singular Gaussian!";
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, Q, Δ, G, Z, e, λ, a, b},
   n = Length@vs;
   Q = Table[q_{vs⟦a⟧,vs⟦b⟧}[L], {a, n}, {b, n}];
   If[(Δ = CF@Det[-Q]) == 0, Message[Integrate::sing]; Return[]];
   Z = CF[L - Sum[Q⟦a, b⟧ θ_{vs⟦a⟧,vs⟦b⟧}, {a, n}, {b, n}] / 2];
   G = Inverse[Q] / 2;
   While[e = D[Z, λ];
    Do[e += G⟦a, b⟧ (D[Z, vs⟦a⟧, vs⟦b⟧] + D[Z, vs⟦a⟧] × D[Z, vs⟦b⟧]), {a, n}, {b, n}];
    e = CF[e];
    ! (e ≡ 0), Z = CF[Z - ∫_0^λ e ⅆλ]
   ];
   PowerExpand@Factor[ω (Δ (2 π)^n)^(-1/2)] × 𝔼[CF[Ev_{vs→0}[Z] /. λ → 1]]
  ];
Protect[Integrate];
```

In[∘]:= $\int \mathbb{E}\left[\mathbb{i}\,\lambda\, x_1^2/2\right] \, ⅆ\{x_1\}$

Out[∘]=

$$\frac{(-1)^{1/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-\mathbb{i}\,\lambda\, x_1^2/2\right] \, ⅆ\{x_1\}$

Out[∘]=

$$-\frac{(-1)^{3/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{\mathbb{i}}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}\right] ⅆ\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{2\sqrt{b^2 - a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda\, x_1^2/2\right] \, ⅆ\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:=
$$\int \mathbb{E}\left[-x_1^2/2+\xi\, x_1\right] \mathrm{d}\{x_1\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{\xi^2}{2}\right]}{\sqrt{2\pi}}$$

In[∘]:=
$$\int \mathbb{E}\left[-\frac{1}{2}\{x_1,x_2\}.\begin{pmatrix} a & b \\ b & c\end{pmatrix}.\{x_1,x_2\}+\{\xi_1,\xi_2\}.\{x_1,x_2\}\right] \mathrm{d}\{x_1,x_2\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2-2\,b\,\xi_1\,\xi_2+a\,\xi_2^2}{2\left(-b^2+a\,c\right)}\right]}{2\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:=
$$\mathrm{I1}=\int \mathbb{E}\left[-\frac{1}{2}\{x_1,x_2\}.\begin{pmatrix} a & b \\ b & c\end{pmatrix}.\{x_1,x_2\}+\{\xi_1,\xi_2\}.\{x_1,x_2\}\right] \mathrm{d}\{x_1\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[-\frac{\left(-b^2+a\,c\right)x_2^2}{2\,a}+\frac{\xi_1^2}{2\,a}+\frac{x_2\left(-b\,\xi_1+a\,\xi_2\right)}{a}\right]}{\sqrt{a}\,\sqrt{2\pi}}$$

In[∘]:=
$$\int \mathrm{I1}\,\mathrm{d}\{x_2\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2-2\,b\,\xi_1\,\xi_2+a\,\xi_2^2}{2\left(-b^2+a\,c\right)}\right]}{2\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:=
$$\int \mathbb{E}\left[\xi\,x+\eta\,y+z\,(x-y)+x^2\right] \mathrm{d}\{x,z\}$$

Out[∘]=
$$-\frac{i\,\mathbb{E}\left[y\,(y+\eta+\xi)\right]}{2\pi}$$

Integration of ϵSeries

In[∘]:=
$$\int \mathbb{E}\left[-x^2/2+\epsilon\,x^3/6+O[\epsilon]^7\right] \mathrm{d}\{x\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\epsilon\mathrm{Series}\left[0,0,\frac{5}{24},0,\frac{5}{16},0,\frac{1105}{1152}\right]\right]}{\sqrt{2\pi}}$$

In[∘]:=
$$\int \mathbb{E}\left[-\phi^2/2+\epsilon\,\phi^4/24+O[\epsilon]^7\right] \mathrm{d}\{\phi\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\epsilon\mathrm{Series}\left[0,\frac{1}{8},\frac{1}{12},\frac{11}{96},\frac{17}{72},\frac{619}{960},\frac{709}{324}\right]\right]}{\sqrt{2\pi}}$$

In[∘]:=
```
∫𝔼[p x + ϵ p^2 x + O[ϵ]^5] ⅆ{p, x}
```

Out[∘]=

$$-\frac{i\,\mathbb{E}[\epsilon\text{Series}[0, 0, 0, 0, 0]]}{2\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
      mon ↦ And[
        Exponent[mon, ϵ] ≤ 2,
        Exponent[mon, x] == Exponent[mon, p]
       ]
     ] &},
  ∫𝔼[p x + a x^2 p + ϵ b x^3 p^3] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,b\,\epsilon+342\,b^2\,\epsilon^2\right]}{2\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
      mon ↦ And[
        Exponent[mon, ϵ] < 4,
        Exponent[mon, x] - Exponent[mon, p] ≤ 3
       ]
     ] &},
  ∫𝔼[p x + a x^2 p + ϵ b p^2 x] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,a\,b\,\epsilon+162\,a^2\,b^2\,\epsilon^2-9072\,a^3\,b^3\,\epsilon^3\right]}{2\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
      mon ↦ And[
        Exponent[mon, ϵ] < 4,
        Exponent[mon, x] - Exponent[mon, p] ≤ 3 - Exponent[mon, ϵ]
       ]
     ] &},
  ∫𝔼[p x + a x^2 p + ϵ b p^2 x] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,a\,b\,\epsilon+162\,a^2\,b^2\,\epsilon^2-9072\,a^3\,b^3\,\epsilon^3\right]}{2\pi}$$

In[ ]:=
```
MatrixForm@Table[
  ∫𝔼[x₁ p₂ + x₂ p₃ + x₃ p₁ + ξᵢ xᵢ + πⱼ pⱼ] ⅆ{x₁, x₂, x₃, p₁, p₂, p₃},
  {i, 3}, {j, 3}]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[-\pi_2\,\xi_1]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} \\ -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[-\pi_3\,\xi_2]}{8\pi^3} \\ -\frac{i\,\mathbb{E}[-\pi_1\,\xi_3]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} \end{pmatrix}$$